
If you're new to the Clipper environment, you may need to take an existing xBase (most any 

dialect past or present) or Clipper Summer '87 release application and give it a facelift by making 

it compatible with the latest version of CA-Clipper. In this article, I'll refer to the current version 

as simply "Clipper." 

 

I, personally, enjoy (and spend a significant part of my consulting practice) improving existing 

applications by changing dated xBase methodologies to more contemporary "Clipper cool" 

techniques. I no longer agree to modify Summer '87 systems without doing a full Clipper 

conversion. The justification for this attitude centers around my belief that future development and 

maintenance will be more streamlined and cost effective once a more robust language comes into 

play--which, in my opinion, justifies the conversion cost up front. 

 

In this article, I explore the ins and outs of such a conversion. You'll need to draw on your basic 

knowledge of current version Clipper language features, programming constructs, and techniques, 

most of which we've covered in past columns. 

 

Levels of conversion 

 

The first thing to consider when you decide to move an existing code base to Clipper, is how 

detailed a conversion is required. There are several ways to proceed: 

 

Enough for an OBJ 

 

This method yields the quickest results, but in the long run, is the least beneficial. Here, you 

use Clipper's compatibility with Summer '87 and dBase III PLUS, along with a few adjustments 

due to a minority of unsupported commands, and come up with a successful compile and resulting 

OBJ file. True, you're now using the most current version of Clipper, but you aren't using any of 

the features that make it a better development platform. 

 

Unobtrusive conversion 

 

Going down this path, you set the stage for a full conversion by using contemporary Clipper 

techniques only where immediately obvious, leaving older methods for another day. 

 

Full conversion 

 

This is the most difficult, but generally yields the most favorable results. Here, you go all the way, 

so to speak, by introducing ragged arrays, built-in classes, code blocks, etc. into the new 

application. The balance of this article focuses on this approach. 

 

Steps toward conversion 

 

To proceed with a seamless, trouble-free evolution to the Clipper language, follow this check list. 

If you pay attention to all these items, you'll achieve a full conversion. I've also thrown in a few 

items that promote an improved coding style, which, when carried forward with the full conversion, 

yields very rewarding results. 

 

Hungarian Notation 

 

A particularly good time to introduce Hungarian notation in an existing application is during the 

conversion. This way you can better manage the expanded number of data types available in 

Clipper, especially when dealing with arrays, code blocks, and objects. 

 

Using Hungarian notation also places necessary limitations on xBase's inherent "dynamic typing" 

facility whereby a variable's data type can be changed by simply assigning it a value of a different 

type. Consider the code segment below: 



 

* In xBase you can define a character variable and  

* then reassign to a numeric.  

 
mZipCode = '90278'  
mZipCode = 90024 && Type re-assignment! 
 

 

* With Hungarian notation, this can t happen if you follow  the rules.  

 
cZipCode = '90278' // 'c' prefix allows only char.  
nZipCode = 90024 // Another memvar is required, thi s time with a 'n' prefix. 
 

You'll find that a lot of legacy xBase and Summer '87 applications use excessive dynamic typing. 

The conversion process is a good time to ferret out the excessive dynamic typing to get a more 

stable code base. 

 

When dealing with Clipper, you should also contemplate explicitly declaring all memvars in LOCAL 
or STATIC statements. This way, there will be one central place to check for a variable's existence 

and its sole data type. 

 

Clipper Operators 

 

Legacy xBase code may need an important facelift with Clipper's series of streamlined operators. 

Here's a list of these operators: 

 

:= assignment  

++, -- pre- and post-increment/decrement  

+=, -=, *=, /=, %=, compound assignment  

== exact equality  

! .NOT != not equal 

 

Introducing these operators during a conversion is important for many reasons. Probably the most 

important of these reasons centers around the ability to embed them as "in-line" expressions in 

logical expressions, parameter lists, and code blocks. 

 

The dated STORE command finds a replacement by the compound assignment operator, yielding a 

more readable form: 

 

* First an Xbase version  

 
STORE O TO nCount1, nCount2, nCount3 
 

* Now in Clipper  

 
nCount1 := nCount2 := nCount3 := 0 
 

Compiling with /W 

 

Whenever compiling a PRG with Clipper, especially during the initial stages of a conversion, I 

recommend using the /W compiler switch. Beware though, as a deluge of warning messages will 

ensue. This tells you how the compiler really feels about your program. Your goal should be to get 

a clean compile (i.e., no errors or warnings). One potentially annoying side affect of Clipper is the 

public array GetList  which is referenced by many commands. This reference generates warning 

messages. You can eliminate the warnings by placing a MEMVAR GetList  statement in each 

program that the compiler diagnoses with a warning. 

 

Shortcutting 

 

The default process of "logical operator shortcutting" in Clipper may leave legacy xBase systems 

running differently than before. Shortcutting involves situations where a logical expression isn't 

evaluated completely (as an optimization technique), causing incorrect results. Consider the 

following example: 

 



IF lDelinq .AND. Fees() < ... 
   Statements > ...  
ENDIF 
 

Here, if the logical value of lDeling is .F., Clipper doesn't evaluate the rest of the expression, 

namely call the Fees() UDF since the value of the expression has been determined because lDelinq 

is false. Unfortunately, Fees() may perform an important calculation. Fees() would have been 

called in other languages, including Summer '87. You can turn off shortcutting for compatibility 

purposes by specifying the /Z compiler switch, but it's recommended during the conversion 

process to re-write the offending logical expressions to take advantage of shortcutting. In our 

example, we see the equivalent code: 

 
LReturn := Fees() // Call Fees() no matter what 
IF lDelinq .AND. lReturn // Now shortcutting OK. 
 

 

TBrowse and TBColumn 

 

One selling point for Clipper is its TBrowse  and TBColumn predefined classes that together provide 

for an advanced, object-based, data browsing technology. For Summer '87 applications, all 

occurrences of DBEDIT()  should be converted (Clipper lists DBEDIT()  as a compatibility function) 

to a TBrowse  equivalent. This may be a difficult process, especially if DBEDIT() 's user function 

parameter was used. Converting from other xBase languages involves taking the BROWSE 
command's functionality and coming up with something similar in Clipper. 

 

The Preprocessor 

 

Wherever appropriate, you should consider using the features of the preprocessor. This includes 

using manifest constants instead of memory variable constants, using the predefined manifest 

constants found in the standard Clipper header files (especially key definitions like K_ESC, K_UP, 

etc.), and introducing pseudo functions to replace simple UDFs. 

 

In the case of logical constants, COMMON.CH contains two very useful constant definitions: 

 
#define TRUE  .T.  
#define FALSE .F. 
 

I've found that TRUE and FALSE make for much more readable logical expressions. 

 

Revised scoping/lifetime 

 

This is one of the most important areas during the conversion process and often the most difficult. 

Many xBase and Summer '87 applications have become extremely tangled due to the misuse of the 

xBase language's ability to "wildly inherit" variables possessing a PRIVATE scope. For example, I've 

seen systems that declare a private memvar, assign it a value, and not refer to it until it reaches a 

module five levels deep in the calling chain. This is an abuse of the inheritance of privates, 

especially if that lower level module is called by other modules. 

 

In Clipper, you should declare regular memvars as LOCAL and pass them to subprograms as 

parameters instead of depending on their inheritance. 

 

There are some instances where limited scale inheritance is acceptable, namely for data-entry 

programs that issue GETs for memvars and call subsidiary modules to scatter and/or gather 

database field contents (although using arrays is more efficient). 

 

During a conversion, you should try to eliminate all PRIVATEs and PUBLICs in favor of the lexically 

scoped LOCAL and STATIC (including file-wide statics). 

 

Remember that a file-wide static give you the same lifetime as a PUBLIC (i.e., forever), but its 

scope is limited to subprograms in the current PRG file. If you truly need "globally scoped" 

variables, those that are known everywhere and live forever, use PUBLICs. Make sure you declare 

PUBLICs in the Main() module 



 

Functionize 

 

Contemporary use of the Clipper language says you should consider "functionizing" all program 

modules. You can group related subprograms in a single PRG file and enclose each module in either 

a FUNCTION or PROCEDURE declaration. You can treat the PROCEDURE form of subprogram as a 

function, calling it as a function. This way the parameter passing techniques are the same for both 

forms of subprograms. The only difference is that a UDF returns a value and a procedure doesn't. 

The most important thing is to get rid of the DO/WITH style of procedure calls. Remember, when 

using this subprogram style, you need the /N compiler switch. 

 

Eliminate macros 

 

There are usually many opportunities to eliminate macros in favor of code blocks. You'll find that 

many older xBase and Summer '87 applications go overboard with the & macro operator. In 

Clipper, you can avoid using macros by opting for extended expressions as in: 

 
LOCAL cDatabase := 'Orders' 
 

* Instead of:  

 
USE &cDatabase  
 

* we can do the following:  

 
USE (cDatabase) // Extended expression  
 
? (cDatabase)->CustNo  //Also with alias notation. 
 

Code blocks also offer an opportunity to avoid macros in certain situations, as in: 

 
LOCAL cSelection := 'Grade > 80’  
LOCAL nTotGrade 
 

* Instead of:  

 
SUM Class->Grade TO nTotGrade FOR &cSelection  
 

* we can do the following:  

 
bSelection := { || Grade > 80 }  
 
SUM Class->Grade TO nTotGrade FOR EVAL (bSelection)  
 

The challenge is to recognize when to substitute a code block for a macro. 

 

Use the enhanced array facilities 

 

Many of the coding decisions and practices that were part of older applications may have been 

based on the limitations or absence of array structures. When converting, these decisions should 

be reconsidered and the ragged array approach applied. In addition, the array processing 

capabilities in the various built-in array functions should be used whenever possible. Be aware, 

however, that introducing arrays to a program that didn't previously use them, normally requires a 

re-write. 

 

The hardest part in applying Clipper's array technology is in recognizing when the program needs 

arrays. Often the need jumps out at you as in: 

 
IF nCounty == 1  
   nTaxAmt := nPrice * nTaxRate1  
ELSEIF nCounty == 2  
   nTaxAmt := nPrice * nTaxRate2  
ELSEIF nCounty == 3  



   nTaxAmt := nPrice * nTaxRate3  
   . .<statements>. 
 

Here, an array can shorten and clarify the code considerably as in: 

 
LOCAL aTaxRate[26] // Array of tax rates  
nTaxAmt := nPrice * aTaxRate[nCounty] 
 

Database functions 

 

A knowledge of Clipper's ability to have the preprocessor translate many database-oriented 

commands to their built-in function equivalents (see STD.CH), sheds light on an alternate way of 

writing Clipper code--calling the database functions directly. Thus, the following Summer '87 code 

might evolve in this manner: 

 

* Summer '87 version  

 
SET SOFTSEEK ON  
mName = 'Gorham'  
SEEK mName  
DO WHILE Customer->Name = mName .AND. .NOT. EOF()  
   ? Customer -> Name, Customer -> Zip  
   SKIP  
ENDDO 
 

/* Clipper version */  

 
LOCAL cName 
DBSEEK(cName, TRUE) // 2nd parm indicates a softsee k  
WHILE Customer->Name == cName .AND. !EOF()  
   QOUT(Customer->Name)  
   QQOUT (Customer->Zip)  
   DBSKIP()  
END 
 

The only gain in calling database functions directly is that they may be aliased as in  

 

Employee->(DBSKIP()) . 

 

Using the order management system 

 

Clipper has a vastly improved system for ordering data in a database file which is based around its 

RDD (replaceable database driver) technology. New concepts such as orders, order bags, and order 

lists, provide compatibility with indexing schemes for dBASE IV, FoxPro, and Paradox. So if the 

Summer '87 system you've been dealing with was bending over backward to get this compatibility, 

the road is now much smoother. 

 

Other worthy mentions 

 

Here are a few miscellaneous items you may consider when doing a Clipper conversion: Get rid of 

obsolete commands such as: 

 
CALL CANCEL CLEAR ALL DIR FIND NOTE RESTORE SCREEN SAVE SCREEN SET COLOR SET EXACT 
SET EXCLUSIVE SET FORMAT SET PROCEDURE SET UNIQUE STORE TEXT/ENDTEXT WAIT 
 

Use the SET()  function to save and restore Clipper settings. For example: 

 

#include 'set.ch' // Save current cursor setting upon entry.  

 
LOCAL LCursorSave := SET(_SET_CURSOR, 0)  
. .<statements>.  
SET(_SET_CURSOR, LCursorSave) // Restore  before le aving. 
 



Convert all FMT screen forms and FRM report forms to native Clipper code. This makes them much 

more flexible for handling future requirements, it also removes the dependence on external files 

(e.g., FMTs and FRMs). 

 

 

In summary 

 

With the tips outlined in this article, you should have a pretty solid idea of what you're up against 

when making the progression to the Clipper 5.2 language. The process may seem arduous at 

times, but the benefits outweigh the difficulties. One last word to convince you a conversion will 

assist future development endeavors is that the coming CA-Visual Objects Windows environment 

contains all the features defined here. Making a transition from Clipper to CA-VO will be all the less 

painful if you start migrating your code to a better way now. 

